
Changelog for v0.x
    


  
    
  
  v0.2.1 (2024-09-23)



  
    
  
  Bug fixes


	Upgrade erlfdb to 0.2.1, allowing Livebook Desktop to discover fdbcli location in /usr/local/bin


  
    
  
  v0.2.0 (2024-09-21)



  
    
  
  Bug fixes



  
    
  
  Enhancements


	Upserts: Support for Ecto options :on_conflict and :conflict_target
	Pipelining: New Repo functions for async/await within a transaction.


  
    
  
  New Documentation


	TESTING.md: Document to describe how to set up a Sandbox
	CHANGELOG.md: This file!



  

    
Testing with EctoFoundationDB
    

Setting up your app to use a FoundationDB Sandbox is very easy!
First, in config/test.exs:
config :my_app, MyApp.Repo,
  open_db: &EctoFoundationDB.Sandbox.open_db/0
Next, set up an ExUnit Case that will provide you with a new Tenant to use
in your tests. This step is optional, but recommended: it will make writing
each test frictionless.
defmodule MyApp.TenantCase do
  @moduledoc false
  use ExUnit.CaseTemplate

  alias EctoFoundationDB.Sandbox

  setup do
    tenant_id = Ecto.UUID.autogenerate()

    tenant = Sandbox.checkout(MyApp.Repo, tenant_id, [])

    on_exit(fn ->
      Sandbox.checkin(MyApp.Repo, tenant_id)
    end)

    {:ok, [tenant_id: tenant_id, tenant: tenant]}
  end
end
Now, you can use your TenantCase to do any FDB operation in a test. Because
we're using tenants with randomized names, you don't have to worry about key conflicts.
defmodule MyAppHelloFDBTest do
  use MyApp.TenantCase

  alias Ecto.Adapters.FoundationDB

  test "hello", context do
    tenant = context[:tenant]

    # An :erlfdb hello world, delete this for your tests
    assert :not_found ==
             FoundationDB.transactional(
               tenant,
               fn tx ->
                 tx
                 |> :erlfdb.get("hello world")
                 |> :erlfdb.wait()
               end
             )

    # Example:
    #   MyApp.Repo.insert!(%MyApp.Hello{message: "world"}, prefix: tenant)
  end
end


  

    
EctoFoundationDB Guide
    

Mix.install([
  {:ecto_foundationdb, "~> 0.1"}
])

  
    
  
  Setup


Hello! This guide simulates what your experience might be when developing an application with EctoFoundationDB. Specifically, it focuses on the mechanism that EctoFoundationDB uses to create and manage indexes.
It assumes the reader is familiar with general Ecto features.
Before we get started, a couple of important points about executing these commands on your system.
If you received an error on the Mix.install setup, please make sure you have both foundationdb-server and foundationdb-clients packages installed on your system. Also, ensure that your Livebook PATH environment variable includes the directory containing the fdbcli binary.

This LiveBook expects your system to have a running instance of FoundationDB, and it writes and deletes data from it. If your system's /etc/foundationdb/fdb.cluster is pointing to a real database, do not execute these commands!

With that out of the way, we'll start off with creating your Repo module.
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app, adapter: Ecto.Adapters.FoundationDB

  use EctoFoundationDB.Migrator

  def migrations() do
    [
      # {0, IndexesMigration}
    ]
  end
end
Notice that the line with IndexesMigration is commented out. We'll come back to this later.

  
    
  
  Developing your app


This next step simulates your app's startup. Normally, you would have a project defining :my_app and the Repo would be included in your supervision tree. In this Guide, we're starting the Repo as an isolated resource.
{:ok, _} = Ecto.Adapters.FoundationDB.ensure_all_started(MyApp.Repo.config(), :temporary)
MyApp.Repo.start_link(log: false)
Next, we define an Ecto.Schema for events that are coming from a temperature sensor. This is a pretty standard Schema module, but notice the usetenant: true option on @schema_context. This ensures that our TemperatureEvents will be written to a Tenant in the database.
defmodule TemperatureEvent do
  use Ecto.Schema

  @schema_context usetenant: true
  @primary_key {:id, :binary_id, autogenerate: true}

  schema "temperature_events" do
    field(:recorded_at, :naive_datetime_usec)
    field(:kelvin, :float)
    field(:site, :string)
    timestamps()
  end
end
We're going to create a module that will help us insert some TemperatureEvents.
defmodule Sensor do
  alias Ecto.Adapters.FoundationDB

  def record(n, tenant) do
    MyApp.Repo.transaction(
      fn ->
        for _ <- 1..n, do: record(nil)
      end,
      prefix: tenant
    )
  end

  def record(tenant) do
    %TemperatureEvent{
      site: "surface",
      kelvin: 373.15 + :rand.normal(0, 5),
      recorded_at: NaiveDateTime.utc_now()
    }
    |> FoundationDB.usetenant(tenant)
    |> MyApp.Repo.insert!()
  end
end
Now, we create and open a new Tenant to store our TemperatureEvents.
alias EctoFoundationDB.Tenant

tenant = Tenant.open!(MyApp.Repo, "experiment-42c")
We're ready to record an event from our temperature sensor! Feel free to Reevaluate this block several times. You'll record 4 new events each time.
for _ <- 1..4, do: Sensor.record(tenant)
We can list all the events from the Tenant. This uses a single FoundationDB Transaction.
MyApp.Repo.all(TemperatureEvent, prefix: tenant)
If there's a large number of events, you can stream them instead of reading them all at once. This uses multiple FoundationDB Transactions.
MyApp.Repo.stream(TemperatureEvent, prefix: tenant)
|> Enum.to_list()
|> length()
Next, we'd like to read all events from "surface". If you're executing this LiveBook in order, you'll receive an exception on this step.
import Ecto.Query

query = from(e in TemperatureEvent, where: e.site == ^"surface")
MyApp.Repo.all(query, prefix: tenant)
Did you get an exception? If so, scroll back up to the defmodule MyApp.Repo block in the Setup section, un-comment the line with IndexesMigration, and Reevaluate that block. Then come back and continue from here. You don't need to Reevaluate other blocks above this text.
👋
Welcome back! You've instructed the Repo to load a migration next time we open a Tenant. But we still need to define that Migration. The block below defines two indexes.
defmodule IndexesMigration do
  use EctoFoundationDB.Migration

  def change() do
    [
      create(index(TemperatureEvent, [:site])),
      create(index(TemperatureEvent, [:recorded_at]))
    ]
  end
end
Now, we re-open the Tenant. Something very important happens here.
This block simulates you restarting your app, and your client reconnecting. We'll just simply call open!/2 again.
tenant = Tenant.open!(MyApp.Repo, "experiment-42c")
Great! If you made it to this step, then the Migration has executed automatically, and the indexes are ready to be used.

  
    
  
  Querying your data


This next block has the same query as the one that threw an exception earlier. This time, you should retrieve the expected events.
import Ecto.Query
query = from(e in TemperatureEvent, where: e.site == ^"surface")
MyApp.Repo.all(query, prefix: tenant)
We can also use the timestamp index that we created in a new query.
now = NaiveDateTime.utc_now()
past = NaiveDateTime.add(now, -1200, :second)

query =
  from(e in TemperatureEvent,
    where: e.recorded_at >= ^past and e.recorded_at < ^now
  )

MyApp.Repo.all(query, prefix: tenant)
Finally, just for fun, let's insert 10,000 TemperatureEvents!
num = 10000
batch = 100

{t, :ok} =
  :timer.tc(fn ->
    Stream.duplicate(batch, div(num, batch))
    |> Task.async_stream(
      Sensor,
      :record,
      [tenant],
      max_concurrency: System.schedulers_online() * 8,
      ordered: false,
      timeout: 30000
    )
    |> Stream.run()
  end)

IO.puts("Done in #{t / 1000} msec")

  
    
  
  Cleaning up


And if you'd like to tidy up, you can easily delete all the data.
# Note: destructive!
MyApp.Repo.delete_all(TemperatureEvent, prefix: tenant)
# Note: destructive!
Tenant.clear_delete!(MyApp.Repo, "experiment-42c")

Adapter module for FoundationDB.
It uses :erlfdb for communicating with the database.

  
    
  
  Installation


Install the latest stable release of FoundationDB from the
official FoundationDB Releases.
The foundationdb-server package is required on any system that will be running
a FoundationDB server instance. For example, it's common to
run the foundationdb-server on your development machine and on managed
instances running a FoundationDB cluster, but not for your stateless Elixir
application server in production.
foundationdb-clients is always required.
Include :ecto_foundationdb in your list of dependencies in mix.exs:
defp deps do
  [
    {:ecto_foundationdb, "~> 0.1"}
  ]
end

  
    
  
  Usage


Define this module in your MyApp application:
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app, adapter: Ecto.Adapters.FoundationDB

  use EctoFoundationDB.Migrator
  def migrations(), do: []
end
Edit your config.exs file to include the following:
config :my_app, MyApp.Repo,
  cluster_file: "/etc/foundationdb/fdb.cluster"
In your app's supervisor, you can start the repo as follows:
children = [
  ...
  MyApp.Repo,
  ...
]

Supervisor.start_link(...)

  
    
  
  Tenants


EctoFoundationDB requires the use of FoundationDB Tenants.
Each tenant you create has a separate keyspace from all others, and a given FoundationDB
Transaction is guaranteed to be isolated to a particular tenant's keyspace.
You'll use the Ecto :prefix option to specify the relevant tenant for each Ecto operation
in your application.
Creating a schema to be used in a tenant.
defmodule User do
  use Ecto.Schema
  @schema_context usetenant: true
  @primary_key {:id, :binary_id, autogenerate: true}
  schema "users" do
    field(:name, :string)
    field(:department, :string)
    timestamps()
  end
  # ...
end
Setting up a tenant.
alias EctoFoundationDB.Tenant

tenant = Tenant.open!(MyApp.Repo, "some-org")
Inserting a new struct.
user = %User{name: "John", department: "Engineering"}
       |> FoundationDB.usetenant(tenant)

user = MyApp.Repo.insert!(user)
Querying for a struct using the primary key.
MyApp.Repo.get!(User, user.id, prefix: tenant)
When a struct is retrieved from a tenant (using :prefix), that struct's metadata
holds onto the tenant reference. This helps to protect your application from a
struct accidentally crossing tenant boundaries due to some unforeseen bug.

  
    
  
  Indexes


The implication of the FoundationDB Layer Concept
is that the manner in which data can be stored and accessed is the responsibility of the client
application. To achieve general purpose data storage and access patterns commonly desired
in web applications, EctoFoundationDB provides support for indexes out of the box, and
also allows your application to define custom indexes.
When you define a migration that calls create index(User, [:department]), a Default index is
created on the :department field. The following two indexes are equivalent:
# ...
create index(User, [:department])
# is equivalent to
create index(User, [:department], options: [indexer: EctoFoundationDB.Indexer.Default]))]
# ...
Timestamp indexes and multi-field indexes are supported:
create index(User, [:inserted_at])
create index(User, [:birthyear, :department, :birthdate])
In fact, the index value can be any Elixir term. All types support Equal queries. However,
certain Ecto Schema types support Between queries. See the Data Types section for the list.
For example, When an index is created on timestamp-like fields, it is an effective time
series index. See the query examples below.
See the Migrations section for further details about managing indexes in your application.

  
    
  
  Index Query Examples


Retrieve all users in the Engineering department (with an Equal constraint):
iex> query = from(u in User, where: u.department == ^"Engineering")
iex> MyApp.Repo.all(query, prefix: tenant)
Retrieve all users with a birthyear from 1992 to 1994 (with a Between constraint):
iex> query = from(u in User,
...>   where: u.birthyear >= ^1992 and u.birthyear <= ^1994
...> )
iex> MyApp.Repo.all(query, prefix: tenant)
Retrieve all Engineers born in August 1992:
iex> query = from(u in User,
...>   where: u.birthyear == ^1992 and
...>          u.department == ^"Engineering" and
...>          u.birthdate >= ^~D[1992-08-01] and u.birthdate < ^~D[1992-09-01]
...> )
iex> MyApp.Repo.all(query, prefix: tenant)
Order matters!: When you create an index using multiple fields, the FDB key that stores the index will be extended with
all the values in the order of your defined index fields. Because FoundationDB stores keys in a well-defined order,
the order of the fields in your index determines the Between queries you can perform.
There can be 0 or 1 Between clauses, and if one exists, it must correspond to the final constraint in the where clause when
compared against the order of the index fields.

  
    
  
  Custom Indexes


As data is inserted, updated, deleted, and queried, the Indexer callbacks (via behaviour EctoFoundationDB.Indexer)
are executed. Your Indexer module does all the work necessary to maintain the index within the transaction
that the data itself is being manipulated. Many different types of indexes are possible. You are in control!
For an example, please see the MaxValue implementation
which is used internally by EctoFoundationDB to keep track of the maximum value of a field in a Schema.

  
    
  
  Transactions


The Repo functions will automatically use FoundationDB transactions behind the scenes. EctoFoundationDB also
exposes the Ecto.Repo transaction function, allowing you to group operations together with transactional isolation:
MyApp.Repo.transaction(fn ->
    # Ecto work
  end, prefix: tenant)
It also exposes a FoundationDB-specific transaction:
alias Ecto.Adapters.FoundationDB

FoundationDB.transactional(tenant,
  fn tx ->
    # :erlfdb work
  end)
Both of these calling conventions create a transaction on FDB. Typically you will use
Repo.transaction/1 when operating on your Ecto.Schema structs. If you wish
to do anything using the lower-level :erlfdb API (rare), you will use
FoundationDB.transactional/2.
Please read the FoundationDB Developer Guid on transactions
for more information about how to develop with transactions.
It's important to remember that even though the database gives ACID guarantees about the
keys updated in a transaction, the Elixir function passed into Repo.transaction/1 or
FoundationDB.transactional/2 will be executed more than once when any conflicts
are encountered. For this reason, your function must not have any side effects other than
the database operations. For example, do not publish any messages to other processes
(such as Phoenix.PubSub) from within the transaction.
# Do not do this!
MyApp.Repo.transaction(fn ->
  MyApp.Repo.insert(%User{name: "John"})
  ...
  Phoenix.PubSub.broadcast("my_topic", "new_user") # Not safe! :(
end, prefix: tenant)
# Instead, do this:
MyApp.Repo.transaction(fn ->
  MyApp.Repo.insert(%User{name: "John"})
  ...
end, prefix: tenant)
Phoenix.PubSub.broadcast("my_topic", "new_user") # Safe :)

  
    
  
  Migrations


At first glance, EctoFoundationDB migrations may look similar to that of :ecto_sql,
but the actual execution of migrations and how your app must be configured are very
different, so please read this section in full.
If your app uses indexes on any of your schemas, you must include a module that implements
the EctoFoundationDB.Migrator behaviour. By default, this is assumed to be your Repo
module, as we have demonstrated above.
Migrations are only used for index management. There are no tables to add, drop, or modify.
When you add a field to your Schema, any read requests on old objects will return nil in
that field.
As tenants are opened during your application runtime, migrations will be executed
automatically. This distributes the migration across a potentially long period of time,
as migrations will not be executed unless the tenant is opened.
The Migrator is a module in your application runtime that provides the full list of
ordered migrations. These are the migrations that will be executed when a tenant is opened.
If you leave out a migration from the list, it will not be applied.
The result of migrations/0 will change over time to include new migrations, like so:
defmodule MyApp.Repo do
  # ...
  def migrations() do
    [
      {0, MyApp.IndexUserByDepartment}
      # At a later date, we my add a new index with a corresponding addition
      # tho the migrations list:
      #    {1, MyApp.IndexUserByRole}
    ]
  end
end
Each migration is contained in a separate module, much like EctoSQL's. However, the operations
to be carried out must be returned as a list. For example, the creation an index
may look like this:
defmodule MyApp.IndexUserByDepartment do
  use EctoFoundationDB.Migration
  def change() do
    [
      create index(User, [:department])
    ]
  end
end
Your migrator and MyApp.<Migration> modules are part of your codebase. They must be
included in your application release.
Migrations can be completed in full at any time with a call to
EctoFoundationDB.Migrator.up_all/1. Depending on how many tenants you have in
your database, and the size of the data for each tenant, this operation might take a long
time and make heavy use of system resources. It is safe to interrupt this operation, as
migrations are performed transactionally. However, if you interrupt a migration, the next
attempt for that tenant may have a brief delay (~5 sec); the migrator must ensure that the
previous migration has indeed stopped executing.

  
    
  
  Data Types


EctoFoundationDB stores your struct's data using :erlang.term_to_binary/1, and
retrieves it with :erlang.binary_to_term/1. As such, there is no data type
conversion between Elixir types and Database Types. Any term you put in your
struct will be stored and later retrieved.
Data types are used by EctoFoundationDB for the creation and querying of indexes.
Certain Ecto types are encoded into the FDB key, which allows you to formulate
Between queries on indexes of these types:
	:id
	:binary_id
	:integer
	:float
	:boolean
	:string
	:binary
	:date
	:time
	:time_usec
	:naive_datetime
	:naive_datetime_usec
	:utc_datetime
	:utc_datetime_usec

See Queries for more information.

  
    
  
  Upserts


The FoundationDB Adapter supports the following upsert approaches.

  
    
  
  Upserts with on_conflict


The following choices for on_conflict are supported, and they work in the manner
described by the offical Ecto documentation.
	:raise: The default.
	:nothing
	:replace_all
	{:replace_all_except, fields}
	{:replace, fields}


  
    
  
  Upserts with conflict_target


If you provide conflict_target: [], the FoundationDB Adapter will make no effort to
inspect the existence of objects in the database before insert. When an existing
object is blindly overwritten, your indexes may become inconsistent, resulting in undefined
behavior. You should only consider its use if one of the following is true:
	(a) your schema has no indexes
	(b) you know apriori that any indexed fields are unchanged at the time of upsert, or
	(c) you know apirori that the primary key you're inserting doesn't already exist.

For these use cases, this option may speed up the loading of initial data at scale.

  
    
  
  Pipelining


Pipelining is the technique of sending multiple queries to a database at the same time, and
then receiving the results at a later time. In doing so, you can often avoid waiting
for multiple network round trips. EctoFDB uses a async/await syntax to
implement pipelining.
All the "queryable" repo functions have async versions:
	Repo.async_get/2 / Repo.async_get/3
	Repo.async_get!/2 / Repo.async_get!/3
	Repo.async_get_by/2 / Repo.async_get_by/3
	Repo.async_get_by!/2 / Repo.async_get_by!/3
	Repo.async_one/1 / Repo.async_one/2
	Repo.async_one!/1 / Repo.async_one!/2
	Repo.async_all/1 / Repo.async_all/2

The results of these functions are provided to Repo.await/1 inside a Repo.transaction/2.
Consider this example. Our transaction below has 2 network round trips in serial.
The "John" User is retrieved, and then the "James" users is retrieved. Finally, the
list is constructed and the transaction ends.
result = MyApp.Repo.transaction(fn ->
  [
    MyApp.Repo.get_by(User, name: "John"),
    MyApp.Repo.get_by(User, name: "James")
  ]
end, prefix: tenant)

[john, james] = result
With pipelining, we can issue 2 async_get_by queries without waiting for their result,
and then only when we need the data, we await the result. Notice that the list of futures
is constructed, and then we await on them.
result = MyApp.Repo.transaction(fn ->
  futures = [
    MyApp.Repo.async_get_by(User, name: "John"),
    MyApp.Repo.async_get_by(User, name: "James")
  ]

  MyApp.Repo.await(futures)
end, prefix: tenant)

[john, james] = result
Conceptually, only 1 network round-trip is required, so this will be faster than
waiting on each in sequence.

  
    
  
  Pipelining on range queries (interleaving waits)


The functions Repo.async_all/1 and Repo.async_all/2 work a little bit different on the await. When the DB must
return a list of results, if that list ends up being larger than :erlfdb's configured :target_bytes,
then the result set is split into multiple round-trips to the database. If your application awaits multiple
such queries, then :erlfdb's "interleaving waits" feature is used.
Consider this example, where we're retrieving all Users and all Posts in the same transaction.
MyApp.Repo.transaction(fn ->
  futures = [
    MyApp.Repo.async_all(User),
    MyApp.Repo.async_all(Post)
  ]

  MyApp.Repo.await(futures)
end, prefix: tenant)
Each query individually executed would be multiple round-trips to the database as pages of results are retrieved.
sequenceDiagram
    Client->>Server: get_range(User)
    Server-->>Client: {100 users, continuation_token}
    Client->>Server: get_range(User, continuation_token)
    Server-->>Client: {10 users, done}
With :erlfdb's interleaving waits, we send as many get_range requests as possible with pipelining
until all are exhausted.
sequenceDiagram
    Client->>Server: get_range(User)
    Client->>Server: get_range(Post)
    Server-->>Client: {100 users, continuation_token}
    Server-->>Client: {100 posts, continuation_token}
    Client->>Server: get_range(User, continuation_token)
    Client->>Server: get_range(Post, continuation_token)
    Server-->>Client: {10 users, done}
    Server-->>Client: {100 posts, continuation_token}
    Client->>Server: get_range(Post, continuation_token)
    Server-->>Client: {2 posts, done}
This happens automatically when you include the list of futures to your Repo.await/1.

  
    
  
  Optimizing for throughput and latency


Pipelining operations inside your transactions can offer significant speed-ups, and you should
also consider parallelizing multiple transactions. Both techniques are useful in different
circumstances.
	Pipelining: The transaction can contain arbitrary logic and remain ACID compliant. If you make
smart use of pipelining, the latency of execution of that logic is kept to a minimum.
	Parallel transactions: Each transaction is internally ACID compliant, but any 2 given
transactions may conflict if they're accessing the same keys. When transactions do not conflict,
throughput goes up.

If you're attempting to get the maximum throughput on data loading, you'll probably want to
make use of both!

  
    
  
  Standard Options


	:cluster_file - The path to the fdb.cluster file. The default is
"/etc/foundationdb/fdb.cluster".
	:migrator - A module that implements the EctoFoundationDB.Migrator
behaviour. Required when using any indexes. Defaults to nil. When nil,
your Repo module is assumed to be the Migrator.


  
    
  
  Advanced Options


	:storage_id - All tenants created by this adapter are prefixed with
this string. This allows multiple configurations to operate on the
same FoundationDB cluster indepedently. Defaults to
"Ecto.Adapters.FoundationDB".
	:open_db - 0-arity function used for opening a reference to the
 FoundationDB cluster. Defaults to :erlfdb.open(cluster_file). When
 using EctoFoundationDB.Sandbox, you should consider setting
 this option to Sandbox.open_db/0.
	:migration_step - The maximum number of keys to process in a single transaction
when running migrations. Defaults to 1000. If you use a number that is
too large, the FDB transactions run by the Migrator will fail.
	:idx_cache - When set to :enabled, the Ecto ets cache is used to store the
available indexes per tenant. This speeds up all database operations.
Defaults to :enabled.


  
    
  
  Limitations and Caveats



  
    
  
  FoundationDB


Please read FoundationDB Known Limitations.

  
    
  
  Layer


EctoFoundationDB implements a specific Layer on the FoundationDB
key-value store. This Layer is intended to be generally useful, but you
may not find it suitable for your workloads. The Layer is documented in
detail at EctoFoundationDB.Layer. This project does not support
plugging in other Layers.

  
    
  
  Tenants


	The use of a tenant implies an ownership relationship between the tenant and the data.
It's up to you how you use this relationship in your application.

	Failure to specify a tenant will result in a raised exception at runtime.



  
    
  
  Migrations


The following are not supported:
	Renaming a table
	Renaming a field
	Dropping an index
	Moving down in migration bersions (rollback)

Finally, the Ecto Mix tasks are known to be unsupported:
# These commands are not supported. Do not use them with :ecto_foundationdb!
#    mix ecto.create
#    mix ecto.drop
#    mix ecto.migrate
#    mix ecto.gen.migration
#    mix ecto.rollback
#    mix ecto.migrations

  
    
  
  Key and Value Size


FoundationDB has limits on key and value size
Please be aware of these limitations as you develop. EctoFoundationDB doesn't make
any attempt to protect you from these errors.

  
    
  
  Ecto Queries


You'll find that most queries outside of those detailed in the documentation fail with
an EctoFoundationDB.Exception.Unsupported exception. The FoundationDB Layer concept
precludes complex queries from being executed within the database. Therefore, it only makes sense
to implement a limited set of query types -- specifically those that do have optimized database
query semantics. All other filtering, aggregation, and grouping must be done by your Elixir code.
In other words, you'll be writing less SQL, and more Elixir.
Refer to the integration tests for a collection of queries that is known to work.

  
    
  
  Other Ecto Features


As you test the boundaries of EctoFoundationDB, you'll find that many other Ecto features just plain
don't work with this adapter. Ecto has a lot of features, and this adapter is in the early stage.
For any given unsupported feature, the underlying reason will be one of the following.
	It makes sense for EctoFoundationDB, but it just hasn't been implemented yet.
	The fundamental differences between FDB and SQL backends mean that there is
no practical reason to implement it in EctoFoundationDB.

Please feel free to request features, but also please be aware that your request
might be closed if it falls into that second category.

See Ecto.Adapters.FoundationDB

EctoFoundationDB.Index 
    



      
See Ecto.Adapters.FoundationDB

Implement this behaviour to create a custom index.

The Ecto FoundationDB Adapter implements a Layer
on the underlying key-value store provided by FoundationDB. Via this layer, some
common data access patterns are achieved. For those familiar with relational databases
such as Postgres, these patterns will be familiar. However, there are many differences (for example SQL is
not supported), so this document seeks to describe the capabilities of the Ecto FoundationDB Layer in detail.

  
    
  
  Keyspace Design


All keys used by :ecto_foundationdb are encoded with FoundationDB's Tuple Encoding
The first element of the tuple is a string prefix that is intended to keep the :ecto_foundationdb keyspace
separate from other keys in the FoundationDB cluster.
Your Schema data and Default indexes are stored with prefix "\xFD".
The data associated with schema migrations is stored with prefix "\xFE".
The rest of the keyspace is open for use by you, the application developer.
All values are either
	other keys (in the case of Default indexes) or
	Erlang term data encoded with :erlang.term_to_binary/1


  
    
  
  Primary Write and Read


Your Ecto Schema has a primary key field, which is usually a string or an integer. This primary
key uniquely identifies an object of your schema within the tenant in which it lives.
defmodule EctoFoundationDB.Schemas.User do
  use Ecto.Schema
  @schema_context usetenant: true
  schema "users" do
    field(:name, :string)
    field(:department, :string)
    timestamps()
  end
end
In this example, a User has an :id and a :name. Also notice that the User is defined with
usetenant: true which provides a scope under which the User lives. For example, a typical tenant
would be the organization the User belongs to. Since the User is in this tenant, we do not need
to provide an identifier for this organization on the User object itself.
"Primary Write" refers to the insertion of the User struct into the FoundationDB key-value store
under a single key that uniquely identifies the User. This key includes the :id value.
Your struct data is stored as a Keyword encoded with :erlang.term_to_binary/1.
Note: The Primary Write can be skipped by providing the write_primary: false option on the @schema_context.
See below for more.

  
    
  
  Default Indexes


When a Default index is created via a migration, the Ecto FoundationDB Adapter writes a set of
keys and values to facilitate lookups based on the indexed field.

  
    
  
  Advanced Options: write_primary: false, mapped?: false


If you choose to use write_primary: false on your schema, this skips the Primary Write. The consequence of this are as follows:
	You'll want to make sure your index is created with an option mapped?: false. This ensures that the
struct data is written to the index keyspace.
	Your index queries will now have performance characteristics similar to a primary key query. That is, you'll
be able to retrieve the struct data with a single erlfdb:get_range/3.
	The data can only be managed by providing a query on the index. You will not be able to access the data
via the primary key.
	If write_primary: false, then only one index can be created.

EctoFoundationDB.Migrator behaviour
    



      
Implement this behaviour to define migrations for Ecto.Adapters.FoundationDB

See Ecto.Adapters.FoundationDB

See Ecto.Adapters.FoundationDB

A module for managing a sandbox FoundationDB cluster. This allows a developer to create
a space under which it should be safe to write tests.
When using this module, it creates a directory in the project root called .erlfdb. It
is safe to delete this directory when you no longer need it (e.g. after test execution)
See Testing with EctoFoundationDB for more.

This module allows the application to manage the creation and deletion of
tenants within the FoundationDB database. All transactions require a tenant,
so any application that uses the Ecto FoundationDB Adapter must use this module.

EctoFoundationDB.Exception.IncorrectTenancy exception
    



      
This exception is raised when there is a conflict in the tenant used.

EctoFoundationDB.Exception.Unsupported exception
    



      
This exception is raised when the application uses an Ecto feature that is not
supported by the FoundationDB Adapter.

